
    Flop Phoenix

Flop Phoenix provides Phoenix components for pagination, sortable tables, and
filter forms with Flop and
Ecto.
Installation
Add flop_phoenix to your list of dependencies in the mix.exs of your Phoenix
application.
def deps do
  [
    {:flop_phoenix, "~> 0.25.2"}
  ]
end
Next, set up your business logic following the
Flop documentation.
Context
Define a context function that performs a list query using Flop.
defmodule MyApp.Pets do
  alias MyApp.Pet

  def list_pets(params) do
    Flop.validate_and_run!(Pet, params, for: Pet, replace_invalid_params: true)
  end
end
Note the usage of the replace_invalid_params option, which lets Flop ignore
invalid parameters instead of producing an error.
LiveView
In the handle_params function of your LiveView module, pass the parameters
to the list function to fetch the data and assign both the data and the meta
struct to the socket.
defmodule MyAppWeb.PetLive.Index do
  use MyAppWeb, :live_view

  alias MyApp.Pets

  @impl Phoenix.LiveView
  def handle_params(params, _, socket) do
    {pets, meta} = Pets.list_pets(params)
    {:noreply, assign(socket, pets: pets, meta: meta)}
  end
end
Sortable table and pagination
Add a sortable table and pagination to your HEEx template:
<Flop.Phoenix.table items={@pets} meta={@meta} path={~p"/pets"}>
  <:col :let={pet} label="Name" field={:name}>{pet.name}</:col>
  <:col :let={pet} label="Age" field={:age}>{pet.age}</:col>
</Flop.Phoenix.table>

<Flop.Phoenix.pagination meta={@meta} path={~p"/pets"} />
The path attribute points to the current route, and Flop Phoenix appends
pagination, filtering, and sorting parameters to it. You can use verified
routes, route helpers, or custom path builder functions. For a
description of the different formats, refer to the documentation of
Flop.Phoenix.build_path/3.
The field attribute in the :col slot is optional. If set and the field
is defined as sortable in the Ecto schema, the table header for
that column will be interactive, allowing users to sort by that column. However,
if the field isn't defined as sortable, or if the field attribute is omitted, or
set to nil or false, the table header will not be clickable.
By using the for option in your Flop query, Flop Phoenix can identify which
table columns are sortable. Additionally, it omits the order and page_size
parameters if they match the default values defined with Flop.Schema.
You also have the option to pass a Phoenix.LiveView.JS command instead of or
in addition to a path. For more details, please refer to the component
documentation.
The pagination component can be used for both page-based pagination and
cursor-based pagination. It chooses the pagination type based on the information
from the Flop.Meta struct.
Event-based pagination and sorting
In the example above, the pagination, sorting, and filtering parameters are
appended to the URL as query parameters. Most of the time, this provides a
better user experience, since users are able to bookmark or share a URL that
leads to the exact view.
In some cases, though, you may prefer not to handle the parameters via the URL.
For example, you may have multiple pageable areas in a single view, or a
pageable widget that is not part of the main content.
In that case, you can set the on_paginate and on_sort attributes instead of
the path attribute and handle these events with the handle_event callback.
Refer to the "Using JS commands" section in the Flop.Phoenix module
documentation for an example.
Filter forms
Flop Phoenix implements the Phoenix.HTML.FormData for the Flop.Meta struct.
As such, you can easily pass the struct to Phoenix form functions. One
straightforward way to render a filter form is through the
Flop.Phoenix.filter_fields/1 component, as shown below:
attr :fields, :list, required: true
attr :meta, Flop.Meta, required: true
attr :id, :string, default: nil
attr :on_change, :string, default: "update-filter"
attr :target, :string, default: nil

def filter_form(%{meta: meta} = assigns) do
  assigns = assign(assigns, form: Phoenix.Component.to_form(meta), meta: nil)

  ~H"""
  <.form
    for={@form}
    id={@id}
    phx-target={@target}
    phx-change={@on_change}
    phx-submit={@on_change}
  >
    <.filter_fields :let={i} form={@form} fields={@fields}>
      <.input
        field={i.field}
        label={i.label}
        type={i.type}
        phx-debounce={120}
        {i.rest}
      />
    </.filter_fields>

    <button name="reset">reset</button>
  </.form>
  """
end
Now you can render a filter form like this:
<.filter_form
  fields={[:name, :email]}
  meta={@meta}
  id="user-filter-form"
/>
You will need to handle the update-filter event with the handle_event/3
callback function of your LiveView.
@impl true
def handle_event("update-filter", params, socket) do
  params = Map.delete(params, "_target")
  {:noreply, push_patch(socket, to: ~p"/pets?#{params}")}
end
Note that while the filter_fields component produces all necessary hidden
inputs, it doesn't automatically render inputs for filter values. Instead, it
passes the necessary details to the inner block, allowing you to customize the
filter inputs with your own input component.
You can pass additional options for each field. Refer to the
Flop.Phoenix.filter_fields/1 documentation for details.
Adding visible inputs for meta parameters
If you want to render visible inputs instead of relying on the hidden inputs
that are automatically added to the form, you can just add them to the form
component:
<.form
  for={@form}
  id={@id}
  phx-target={@target}
  phx-change={@on_change}
  phx-submit={@on_change}
>
  <%!-- ... --%>

  <label for="filter-form-page-size">Page size</label>
  <input
    id="filter-form-page-size"
    type="text"
    name="page_size"
    value={@meta.page_size}
  />

  <button name="reset">reset</button>
</.form>
Phoenix.LiveView.JS command as an attribute to the components in that case.
LiveView streams
To use LiveView streams, you can change your handle_params/3 function as
follows:
def handle_params(params, _, socket) do
  {pets, meta} = Pets.list_pets(params)
  {:noreply, socket |> assign(:meta, meta) |> stream(:pets, pets, reset: true)}
end
When using LiveView streams, you need to pass @streams.pets instead of @pets
to the table component.
The stream values are tuples, with the DOM ID as the first element and the items
(in this case, Pets) as the second element. You need to match on these tuples
within the :let attributes of the table component.
<Flop.Phoenix.table items={@streams.pets} meta={@meta} path={~p"/pets"}>
  <:col :let={{_, pet}} label="Name" field={:name}>{pet.name}</:col>
  <:col :let={{_, pet}} label="Age" field={:age}>{pet.age}</:col>
</Flop.Phoenix.table>
Customization
For customizing the components, it is recommend to define wrapper components
that set the necessary attributes. Refer to the
module documentation for examples.


  

    Changelog

Unreleased
[0.25.2] - 2025-06-25
Fixed
	Allow overriding maxlength attribute in filter value field.

[0.25.1] - 2025-05-05
Fixed
	Function clause error when no pagination parameters are passed and no
default_pagination_type is set.

[0.25.0] - 2025-05-04
Added
	Add guide about load-more buttons and infinite scroll.
	Add guide about CSS styles.
	Add guide about page size controls.

Changed
	Remove opts attribute from Flop.Phoenix.pagination/1 and remove
configuration of the pagination component via application environment.
	Remove all default classes from pagination component.
	Remove wrapper_attrs from pagination component in favor of global
attributes.
	Set page link aria label function with page_link_aria_label_fun attribute
on Flop.Phoenix.pagination/1 component instead of using
pagination_link_aria_label option.
	Set previous and next link attributes and content with slots instead of
options.
	Set ellipsis content with slot instead of option.
	Set pagination list attributes with page_list_attrs attribute instead of
pagination_list_attrs option.
	Set pagination list item attributes with page_list_item_attrs attribute
instead of pagination_list_item_attrs option.
	Set pagination link attributes with page_link_attrs attribute
instead of pagination_link_attrs option.
	Set current pagination link attributes with current_page_link_attrs
attribute instead of current_link_attrs option.
	Set disabled link attributes with disabled_link_attrs attribute instead of
disabled_attrs option.
	Remove default classes for pagination root element, pagination list and
pagination links.
	Use <button> elements for pagination if no path is set.
	Remove role attribute from the pagination component. The <nav> element
already has the implicit ARIA role navigation.
	Add rel attribute to previous/next links.
	Mark up disabled previous/next links of the pagination component as
<a role="link" aria-disabled="true"> instead of <span>.
	Update documentation for hidden_inputs_for_filter/1 to use
Phoenix.Component.inputs_for/1 with the skip_persistent_id option.
	Require Phoenix >= 1.6.0 and < 1.9.0.

How to upgrade
The configuration via pagination_opts is deprecated. Instead of referencing a
function that returns the pagination opts, it is recommended to define a
component that wraps the Flop.Phoenix.pagination/1 component and sets the
necessary attributes.
Follow the upgrade guide of version 0.24.0 to update the pagination component
configuration, including the removal of the application configuration, if you
haven't done so already.
Remove the wrapper_attrs from your pagination options and pass them directly
as attributes instead.
  <Flop.Phoenix.pagination
    meta={@meta}
    path={@path}
    opts={[
-     wrapper_attrs: [
-       class: "pagination",
-       aria: [label: "Quppernerit"]
-     ]
    ]}
+   class="pagination"
+   aria-label="Quppernerit"
  />
Replace the :previous_link_attrs, :previous_link_content, next_link_attrs,
and next_link_content attributes with the previous and next slots.
  <Flop.Phoenix.pagination
    meta={@meta}
    path={@path}
    opts={[
-     previous_link_attrs: [class: "previous"],
-     previous_link_content: Phoenix.HTML.raw(~s(<i class="fas fa-chevron-left" />)),
-     next_link_attrs: [class: "next"],
-     next_link_content: Phoenix.HTML.raw(~s(<i class="fas fa-chevron-right" />))
    ]}
- />
+ >
+   <:previous attrs={[class: "previous"]}>
+     <i class="fas fa-chevron-left" />
+   </:previous>
+   <:next attrs={[class: "next"]}>
+     <i class="fas fa-chevron-right" />
+   </:next>
+ </Flop.Phoenix.pagination>
Replace the :ellipsis_attrs and :ellipsis_content attributes with the
ellipsis slot.
  <Flop.Phoenix.pagination
    meta={@meta}
    path={@path}
    opts={[
-     ellipsis_attrs: [class: "ellipsis", aria-hidden: "true"],
-     ellipsis_content: "‥"
    ]}
- />
+ >
+   <:ellipsis>
+     <span class="ellipsis" aria-hidden="true">‥</span>
+   </:ellipsis>
+ </Flop.Phoenix.pagination>
Remove the :disabled_attrs option. Note that the disabled and
aria-disabled attributes are set automatically and do not need to be passed
here.
To style disabled links/buttons without a CSS class, use the CSS selector
[disabled], [aria-disabled="true"].
  <Flop.Phoenix.pagination
    meta={@meta}
    path={@path}
    opts={[
-     disabled_attrs: [class: "is-disabled"]
    ]}
+     disabled_link_attrs={[class: "is-disabled"]}
  >
Remove the :pagination_link_aria_label option and set the
page_link_aria_label_fun attribute instead.
  <Flop.Phoenix.pagination
    meta={@meta}
    path={@path}
    opts={[
-     pagination_link_aria_label: &"#{&1}ページ目へ"
    ]}
+     page_link_aria_label_fun={&"#{&1}ページ目へ"}
  >
Remove the :pagination_list_attrs, :pagination_list_item_attrs,
:pagination_link_attrs, and :current_link_attrs options and set the
page_list_attrs, page_list_item_attrs, page_link_attrs, and
current_page_link_attrs attributes instead. If you relied on the default page
list class and page link classes, set them explicitly. Note that the
aria-current attribute is set automatically and does not need to be set here.
To style the current page link without a CSS class, use the CSS selector
[aria-current="page"].
  <Flop.Phoenix.pagination
    meta={@meta}
    path={@path}
    opts={[
-     pagination_list_attrs: [class: "pagination-list"]
-     pagination_list_item_attrs: [class: "pagination-list-item"]
-     pagination_link_attrs: [class: "pagination-link"]
-     current_link_attrs: [
-       class: "pagination-link is-current",
-       aria: [current: "page"]
-     ]
    ]}
+     page_list_attrs={[class: "pagination-list"]}
+     page_list_item_attrs={[class: "pagination-list-item"]}
+     page_link_attrs={[class: "pagination-link"]}
+     current_page_link_attrs={[class: "pagination-link is-current"]}
  >
Ensure that your CSS styles work with both <a> elements (used when a path is
set) and <button> elements (used when no path is set). Check the appearance
of the pagination component carefully.
[0.24.0] - 2025-02-01
Added
	Add Flop.Phoenix.Pagination struct to hold information needed to render
a pagination component.
	Add Flop.Phoenix.Pagination.new/2 to build a Pagination struct from a
Flop.Meta struct.
	Add Flop.Phoenix.pagination_for/1 for building custom pagination components.
Update the existing Flop.Phoenix.pagination/1 component to use it.
	Add Flop.Phoenix.page_link_range/3 for determining which page links to
render in a pagination component.

Changed
	Support cursor pagination in Flop.Phoenix.pagination/1.
	Remove the page_links option from the pagination_opts. Set this option
with the new page_links attribute instead.
	Change the values of the page_links option from
:all | :hide | {:ellipsis, pos_integer} to :all | :none | pos_integer.
	Change default value for page_links option of the
Flop.Phoenix.pagination/1 component from :all to 5.
	Deprecate configuring the pagination and table components via the
application environment. Define wrapper components that pass the opts
attribute to the Flop.Phoenix components instead.
	Remove dependency on PhoenixHTMLHelpers.
	Require Phoenix LiveView ~> 1.0.0.
	Require Elixir ~> 1.14.

Removed
	Remove Flop.Phoenix.cursor_pagination/1. Use Flop.Phoenix.pagination/1
instead.
	Remove t:Flop.Phoenix.cursor_pagination_option/0.
	Remove Flop.Phoenix.IncorrectPaginationTypeError.
	Remove input_type/3 from the Phoenix.HTML.FormData protocol
implementation for Flop.Meta. The function had been removed from the
protocol in Phoenix.HTML 4.0.
	Remove the previously deprecated event attribute from the
Flop.Phoenix.pagination/1 and Flop.Phoenix.table/1 components. Use
:on_paginate and :on_sort instead.
	Remove the previously deprecated hide and show attributes from the
:col and :action slots of the Flop.Phoenix.table/1 component. Use the
:if attribute instead.

Fixed
	Fix a warning about ranges in Elixir 1.18.

How to upgrade
Replace Flop.Phoenix.cursor_pagination/1 with Flop.Phoenix.pagination/1.
- <Flop.Phoenix.cursor_pagination meta={@meta} path={~p"/pets"} />
+ <Flop.Phoenix.pagination meta={@meta} path={~p"/pets"} />
Update the format of the page_links option for the pagination component.
- page_links: {:ellipsis, 7},
+ page_links: 7,

- page_links: :hide,
+ page_links: :none,
Remove page_links from your pagination_opts function and add it as an
attribute instead.
def pagination_opts do
  [
    ellipsis_attrs: [class: "ellipsis"],
    ellipsis_content: "‥",
    next_link_attrs: [class: "next"],
    next_link_content: next_icon(),
-     page_links: 7,
    pagination_link_aria_label: &"#{&1}ページ目へ",
    previous_link_attrs: [class: "prev"],
    previous_link_content: previous_icon()
  ]
end

<Flop.Phoenix.pagination
  meta={@meta}
  path={~p"/pets"}
+ page_links={7}
/>
Replace the :show and :hide attribute in the :col slot of the table
component with :if.
<:col
  :let={pet}
- show={@admin?}
+ :if={@admin?}
  label="Name"
  field={:name}
>
  <%= pet.name %>
</:col>

<:col
  :let={pet}
- hide={!@admin?}
+ :if={@admin?}
  label="Name"
  field={:name}
>
  <%= pet.name %>
</:col>
Replace the event attribute of the pagination table components with
on_paginate and on_sort.
<Flop.Phoenix.pagination
  meta={@meta}
- event="paginate"
+ on_paginate={JS.push("paginate")}
/>

<Flop.Phoenix.table
  items={@pets}
  meta={@meta}
- event="sort"
+ on_sort={JS.push("sort")}
>
Remove the configuration for the pagination component from config/config.exs
and define a wrapper component in your CoreComponents module instead. This
is optional, but will make future version updates easier.
For the pagination component:
# config/config.exs

config :flop_phoenix,
- pagination: [opts: {MyAppWeb.CoreComponents, :pagination_opts}],
  table: [opts: {MyAppWeb.CoreComponents, :table_opts}]

# MyAppWeb.CoreComponents

- def pagination_opts do
-   [
-     # ...
-   ]
- end

+ attr :meta, Flop.Meta, required: true
+ attr :path, :any, default: nil
+ attr :on_paginate, JS, default: nil
+ attr :target, :string, default: nil
+
+ def pagination(assigns) do
+   ~H\"""
+   <Flop.Phoenix.pagination
+     meta={@meta}
+     path={@path}
+     on_paginate={@on_paginate}
+     target={@target}
+     opts={[
+       # ...
+     ]}
+   />
+   \"""
+ end
[0.23.1] - 2024-10-17
Changed
	Raise an error if a meta struct with the wrong pagination type is passed to
the pagination or cursor_pagination component.

Fixed
	Fix compilation error in Phoenix LiveView 1.0.0-rc.7.
	Fix type of row_click attribute.

[0.23.0] - 2024-08-18
Changed
	Support and require live_view ~> 1.0.0-rc.0.
	Allow to pass options directly in config file instead of referencing function.

Fixed
	Fixed a deprecation warning in Elixir 1.17.

[0.22.10] - 2024-08-18
Changed
	Loosen version requirement for flop to support 0.26.

[0.22.9] - 2024-05-04
Added
	Added :pagination_list_item_attrs option to Flop.Phoenix.pagination/1.

[0.22.8] - 2024-03-23
Added
	Support :col_class attr on :col and :action slots in addition to
:col_style.

Changed
	Don't render empty style attributes on col elements in colgroup.

Fixed
	The page range calculation in the Flop.Phoenix.pagination/1 was incorrect
towards the last pages.

[0.22.7] - 2024-03-02
Changed
	Loosen version requirement for phoenix_html.

Fixed
	Warning when wrapping table component and passing on :col slot as attribute.

[0.22.6] - 2024-01-14
Changed
	Support Flop 0.25.0.
	Update documentation examples for filter forms.

[0.22.5] - 2023-12-24
Changed
	Requires phoenix_html ~> 4.0.

[0.22.4] - 2023-11-18
Fixed
	Don't render li element if a pagination link is not rendered.

[0.22.3] - 2023-11-14
Changed
	Support Flop ~> 0.24.0.

[0.22.2] - 2023-10-19
Fixed
	Numbered pagination links were not wrapped in li elements.

[0.22.1] - 2023-09-28
Changed
	Allow to use Phoenix.HTML.safe/0 as a label attribute in table headers.

[0.22.0] - 2023-09-26
Added
	Added directions attribute to the col slot of the table component. This
allows you to override the default sort directions, e.g. to specify
nulls-first or nulls-last.
	Added thead_th_attrs and th_wrapper_attrs attributes to the col slot
of the table component.
	Added thead_th_attrs attribute to the action slot of the table component.

Changed
	Renamed attrs attribute on the col and action slots of the table
component to tbody_td_attrs in order to match the naming of the global
table options.

How to upgrade
Rename the attrs attribute to tbody_td_attrs in both the col slot and the
action slot:
<Flop.Phoenix.table items={@pets} meta={@meta} path={~p"/pets"}>
-  <:col :let={p} attrs={[class="my-class"]}><%= p.id %></:col>
-  <:action :let={p} attrs={[class="my-class"]}>button</:col>
+  <:col :let={p} tbody_td_attrs={[class="my-class"]}><%= p.id %></:col>
+  <:action :let={p} tbody_td_attrs={[class="my-class"]}>button</:col>
</Flop.Phoenix.table>
[0.21.2] - 2023-09-26
Changed
	Support phoenix_liveview ~> 0.20.

[0.21.1] - 2023-08-05
Added
	Allow passing a function to the attrs option of table component's :action
slot. Before, this was only supported in the :col slot.

Changed
	Improve some error messages and documentation examples.

Fixed
	In the :col and :action slots of the table component, the attrs option
did not properly override the attributes set with the :tbody_td_attrs
option.

[0.21.0] - 2023-07-17
Changed
	Depend on flop ~> 0.22.0.

[0.20.0] - 2023-07-09
Added
	Added an on_paginate attribute to the pagination and cursor_pagination
components. This attribute takes a Phoenix.LiveView.JS command as a value.
The attribute can be combined with the path attribute, in which case the URL
will be patched and the the given JS command is executed.
	Similarly, an on_sort attribute was added to the table component.
	Allow setting tbody_tr_attrs to a function that takes the current row item
as an argument and returns a list of attributes.
	Allow setting the attrs attribute of the :col slot of the table component
to a function that takes the current row item as an argument and returns a
list of attributes.

Changed
	The table ID falls back to "sortable-table" if no schema module is used.
	The tbody ID was changed to {id}-tbody.
	The table container ID is set to {id}-container now.
	The table ID is set to {id} now.

Deprecated
	The show and hide attributes of the :col slot of the table component are
now deprecated in favor of the :if attribute.
	The event attribute of the pagination, cursor pagination and table
components has been deprecated in favor of on_pagination and on_sort.

[0.19.1] - 2023-06-30
Changed
	The table component only renders an ascending/descending order indicator in
the column header for the first order field, instead rendering one in the
column headers for all ordered fields.
	Support Flop 0.22.

[0.19.0] - 2023-05-30
Changed
	Necessary updates for phoenix_live_view ~> 0.19.0.
	Requires phoenix_live_view ~> 0.19.0.
	Remove previously deprecated Flop.Phoenix.pop_filter/2. Use
Flop.Filter.pop/3 instead.

[0.18.2] - 2023-05-08
Changed
	Added hidden attribute to hidden inputs rendered by filter_fields
component in order to solve CSS spacing issues.

[0.18.1] - 2023-03-18
Changed
	Added support for LiveView streams to the table component. To this end,
row_id, row_item and id attributes were added to the component,
following the example of Phoenix 1.7. The id attribute is added to the
tbody. If no id is explicitly set, a default value will be used depending
on the schema name.

[0.18.0] - 2023-02-25
Changed
	Added dependency on Phoenix.HTML ~> 3.3.0.
	The filter_fields component now passes the Phoenix.HTML.FormField struct
introduced in Phoenix.HTML 3.3.0 to the inner block.
	Support :as option for filter inputs with Phoenix.HTML.FormData/4
(inputs_for).
	Phoenix.HTML.FormData.input_type/2 now considers the Ecto type for join,
custom and compound fields.
	Remove support for path_helper assigns, previously deprecated in 0.15.
	Deprecate Flop.Phoenix.pop_filter/2.

How to upgrade
If your input component already knows how to handle the
Phoenix.HTML.FormField struct, you can update the inner block for
filter_fields like this:
<.filter_fields :let={i} form={f} fields={[:email, :name]}>
  <.input
-    field={{i.form, i.field}}
+    field={i.field}
    type={i.type}
    label={i.label}
-    id={i.id}
-    name={i.name}
-    value={i.value}
    {i.rest}
  />
</.filter_fields>
If your input component still expects the individual assigns, you can update
the inner block like this:
<.filter_fields :let={i} form={f} fields={[:email, :name]}>
  <.input
-    field={{i.form, i.field}}
+    field={{i.field.form, i.field.field}}
    type={i.type}
    label={i.label}
-    id={i.id}
-    name={i.name}
-    value={i.value}
+    id={i.field.id}
+    name={i.field.name}
+    value={i.field.value}
    {i.rest}
  />
</.filter_fields>
For an upgrade example for the path_helper assign, see the changelog entry for
version 0.15.0.
[0.17.2] - 2023-01-15
Changed
	Support Flop 0.19.

[0.17.1] - 2022-11-15
Added
	Allow passing an offset when generating filter inputs with
Phoenix.HTML.Form.inputs_for/3.

[0.17.0] - 2022-10-27
Added
	New option tbody_attrs for table component.
	New attribute row_click and slot action for table component.

Changed
	To pass additional attributes to a column, you will now have to use the
attrs attribute. This was necessary because defining a global attribute on
a slot causes a compile-time error in phoenix_live_view 0.18.3.

<Flop.Phoenix.table items={@pets} meta={@meta} path={~p"/pets"}>
-  <:col :let={p} class="my-class"><%= p.id %></:col>
+  <:col :let={p} attrs={[class="my-class"]}><%= p.id %></:col>
</Flop.Phoenix.table>
[0.16.0] - 2022-10-10
Added
	New Phoenix component Flop.Phoenix.hidden_inputs_for_filter/1.

Changed
	Major refactoring of Flop.Phoenix.filter_fields/1. Instead of giving you the
rendered <label> and <input> elements, the component now only passes the
necessary arguments to the inner block. You will have to pass these arguments
to your own input component (or whatever you name it). The field option
format has also been updated. These changes were made to fix warnings emitted
by live view 0.18.1, and also accompany current changes in Phoenix that thin
out Phoenix.HTML / Phoenix.HTML.Form in favor of Phoenix components for
inputs.
	Require flop >= 0.17.1 and < 0.19.0.

Removed
	Removed Flop.Phoenix.filter_hidden_inputs_for/1. This function is not used
internally anymore. You can either use Phoenix.HTML.Form.hidden_inputs_for/1
(Phoenix.HTML ~> 3.2), or use Flop.Phoenix.hidden_inputs_for_filter/1,
which does the same, but as a Phoenix component.
	Removed Flop.Phoenix.filter_label/1 and Flop.Phoenix.filter_input/1. With
the changes to Flop.Phoenix.filter_fields/1 and the move away from the
input rendering functions of Phoenix.HTML.Form, these functions don't have
any value anymore. Read the documentation of
Flop.Phoenix.hidden_inputs_for_filter/1 for an example on how to easily
render the fields of individual filters.

Fixed
	Fixed warnings about tainted variables in live view 0.18.1.
	Fixed an issue where default parameters set in the backend module were not
removed from the query parameters.
	Fixed URLs ending with ? when no query parameters are necessary if the path
is passed as a string.

How to upgrade
Filter fields component
Previously, you would render a filter form like this:
<.form :let={f} for={@meta}>
  <Flop.Phoenix.filter_fields :let={entry} form={f} fields={[:name, :email]}>
    <div class="field">
      <%= entry.label %>
      <%= entry.input %>
    </div>
  </Flop.Phoenix.filter_fields>
</.form>
In this example, entry.label and entry.input are complete <label> and
<input> elements with all attributes set. You will need to change this to:
<.form :let={f} for={@meta}>
  <.filter_fields :let={i} form={f} fields={[:name, :email]}>
    <.input
      id={i.id}
      name={i.name}
      label={i.label}
      type={i.type}
      value={i.value}
      field={{i.form, i.field}}
      {i.rest}
    />
  </.filter_fields>
</.form>
You will have to define an input component in your project. You can take a
hint from the input component that is generated as part of the Components
module by Phoenix 1.7.
Field options
Remove input_opts and label_opts and pass them directly to your input
component, or add them directly to the input component. If you passed an id
to filter_fields, set in on the form instead.
<.filter_fields
  :let={i}
  form={f}
  fields={[:name]}
-  id="some-id"
-  input_opts={[class: "input", phx_debounce: 100]}
-  label_opts={[class: "label"]}
>
  <.input
     ...
+    class="input"
+    phx-debounce={100}
  />
</.filter_fields>
Use strings instead of atoms to set the type, and use the types that your
input component understands.
<.filter_fields
  :let={i}
  form={f}
  fields={[
-    name: [type: :text_input],
+    name: [type: "text"],
-    age: [type: :number_input],
+    age: [type: "number"],
-    phone: [type: :telephone_input],
+    phone: [type: "tel"]
  ]}
>
If you passed additional input function options in a tuple, take them out of
the tuple and add them to the keyword list instead.
<.filter_fields
  :let={i}
  form={f}
  fields={[
-    role: [type: {:select, ["author", "editor"], class: "select"}]
+    role: [type: "select", options: ["author", "editor"], class: "select"]
  ]}
>
The default option is not handled for you anymore. You can still set it, but
it will just be passed on as part of the rest options, so your input
component will need to handle it.
Filter label and input components
If you used Flop.Phoenix.filter_label/1 or Flop.Phoenix.filter_input/1
before, follow the example in the documentation of
Flop.hidden_inputs_for_filter/1 to render the inputs of individual filters
without the removed components.
[0.15.2] - 2022-10-10
Changed
	Change version requirement for Flop to >= 0.15.0 and < 0.19.0.

[0.15.1] - 2022-09-30
Fixed
	Typespec of Flop.Phoenix.build_path/3.

[0.15.0] - 2022-09-22
Added
This release adds support for passing URI strings instead of MFA or FA tuples to
components and helper functions. This allows you to use the library with the
verified routes introduced in Phoenix 1.7. Alternatively, you can now also
choose to pass a 1-ary path builder function. See Flop.Phoenix.build_path/3
for examples. Passing tuples pointing to route helpers is still supported.
	Added an example for a custom filter form component to the readme.
	Support passing a URI string as a path to the table, pagination and
cursor_pagination components and to build_path/3.
	Support passing a 1-ary path builder function to the table, pagination and
cursor_pagination components and to build_path/3.
	New function Flop.Phoenix.pop_filter/2.

Changed
	Require live_view ~> 0.18.0.
	Deprecate path_helper assign in favor of path.
	Use declarative assigns and replace Phoenix.LiveView.Helpers.live_patch/1
with Phoenix.Component.link/1.
	Flop.Phoenix.filter_input/1 requires additional options for the input
function to be passed in the input_opts assign, instead of passing them
directly to the component. This was necessary because the global attributes
you can define with declarative assigns in LiveView 0.18 are meant for HTML
attributes, while the input options may contain any additional attributes
necessary (e.g. a list of select options that are rendered as option
elements).

Fixed
	Apply show and hide attribute for columns to colgroup as well.
	Correctly handle multiple inputs for the same field in Flop.filter_fields/1.

How to upgrade
Rename the path_helper assigns of table, pagination and
cursor_pagination components to path.
- <.pagination meta={@meta} path_helper={{Routes, :pet_path, [@socket, :index]}} />
+ <.pagination meta={@meta} path={{Routes, :pet_path, [@socket, :index]}} />
Wrap additional options passed to Flop.Phoenix.filter_input/1 into a single
input_opts assign.
- <.filter_input form={ff} class="select" options={[:some, :options]} />
+ <.filter_input form={ff} input_opts={[class: "select", options: [:some, :options]]} />
[0.14.2] - 2022-08-26
Changed
	Support Flop ~> 0.17.0.

[0.14.1] - 2022-03-22
Changed
	Support Flop ~> 0.16.0.

[0.14.0] - 2022-02-22
Added
	symbol_unsorted option for the table component.
	caption assign for the table component.
	col_style assign for the :col slot of the table component.

Changed
	Additional attributes passed to the <:col> slot will now be added as
attributes to the <td> tags.

[0.13.0] - 2021-11-14
Added
	Add cursor_pagination/1 component.

Changed
	The pagination component adds the disabled class to the span that is
displayed when the previous or next link is disabled now. Previously, the
disabled attribute was set on the span. The class can be customized with
the :disabled_class option.

[0.12.0] - 2021-11-08
Added
	Implement the Phoenix.HTML.FormData protocol for Flop.Meta. This means
you can pass the meta struct as :for option to the Phoenix form_for
functions now.
	Add the functions filter_fields/1, filter_input/1 and filter_label/1.

Changed
	Remove :for option. The schema module is now automatically derived from the
meta struct.

[0.11.1] - 2021-10-31
Added
	Adds hide and show options to table :col.

Changed
	Passing a label to a table :col is now optional.

[0.11.0] - 2021-10-30
Changed
	The path_helper_args assign has been removed in favor of passing mfa
tuples as path_helper.
	In the same vein, Flop.Phoenix.build_path/4 has been replaced with
Flop.Phoenix.build_path/3, which also takes a tuple as the first argument.
	The table component has been changed to use slots. The headers,
footer, row_func and row_opts assigns have been removed. Also, the
tfoot_td_attrs and tfoot_th_attrs options have been removed.
	The live_view version requirement has been changed to ~> 0.17.0.
	Better error messages for invalid assigns have been added.

How to upgrade
Update the path_helper and path_helper_args assigns set for the table
and pagination component:
- path_helper={&Routes.pet_path/3}
- path_helper_args={[@socket, :index]}
+ path_helper={{Routes, :pet_path, [@socket, :index]}}
If you prefer, you can pass a function instead.
+ path_helper={{&Routes.pet_path/3, [@socket, :index]}}
Update any calls to Flop.Phoenix.build_path/4:
- Flop.Phoenix.build_path(&Routes.pet_path/3, [@socket, :index], meta)
+ Flop.Phoenix.build_path({Routes, :pet_path, [@socket, :index]}, meta)
If you prefer, you can use a 2-tuple here as well:
+ Flop.Phoenix.build_path({&Routes.pet_path/3, [@socket, :index]}, meta)
Finally, update the tables in your templates:
<Flop.Phoenix.table
  for={MyApp.Pet}
  items={@pets}
  meta={@meta}
-   path_helper={&Routes.pet_path/3}
-   path_helper_args={[@socket, :index]}
+   path_helper={{Routes, :pet_path, [@socket, :index]}}
-   headers={[{"Name", :name}, {"Age", :age}]}
-   row_func={fn pet, \_opts -> [pet.name, pet.age] end}
-   footer={["", @average_age]}
- />
+ >
+   <:col let={pet} label="Name" field={:name}><%= pet.name %></:col>
+   <:col let={pet} label="Age" field={:age}><%= pet.age %></:col>

+   <:foot>
+     <tr>
+       <td></td>
+       <td><%= @average_age %></td>
+     </tr>
+   </:foot>
+ </Flop.Phoenix.table>
Also, you can remove tfoot_td_attrs and tfoot_th_attrs from the opts
assign (or opts provider function).
[0.10.0] - 2021-10-24
Added
	It is now possible to set global options for the components in your config.

config :flop_phoenix,
  pagination: [opts: {MyApp.ViewHelpers, :pagination_opts}],
  table: [opts: {MyApp.ViewHelpers, :table_opts}]
Changed
	The for, event and target options moved from the opts assign to the
root. The opts assign is now exclusively used for customization options
that modify the appearance, which are usually set globally for a
project and are not related to the specific data or view.
	The row_func/2 function passed to the table component receives the new
row_opts assign now instead of the opts assign.
	The pagination and table components only pass the for option to the query
builder, instead of all opts.
	The path_helper and path_helper_args assigns are now optional if an
event is passed. A descriptive error is raised if neither of them are
passed.
	The opts assign for the pagination and table components is now optional.
	Aria labels were added to the links to the first and last page.
	The aria-sort attribute was added to the table headers.

How to upgrade
	Remove the for, event and target from the opts assign and add them
as regular assigns at the root level.
	Move any key/value pairs that are needed by your row_func from opts to
row_opts.

For example, if your row_func looks like this:
def table_row(%Pet{id: id, name: name}, opts) do
  socket = Keyword.fetch!(opts, :socket)
  [name, link("show", to: Routes.pet_path(socket, :show, id))]
end
Update your template like this:
<Flop.Phoenix.sortable_table
  row_func={&table_row/2}
-   opts={[
-     container: true,
-     for: Pet,
-     event: "sort-table",
-     target: @myself,
-     socket: @socket
-   ]}
+   row_opts={[socket: @socket]}
+   for={Pet}
+   event="sort-table"
+   target={@myself}
+   opts={[container: true]}
/>

<Flop.Phoenix.pagination
-   opts={[
-     for: Pet,
-     event: "paginate",
-     target: @myself,
-     page_links: {:ellipsis, 7}
-   ]}
+   for={Pet}
+   event="paginate"
+   target={@myself}
+   opts={[page_links: {:ellipsis, 7}]}
/>
[0.9.1] - 2021-10-22
Changed
	Change live_view version requirement to ~> 0.16.0 or ~> 0.17.0.

[0.9.0] - 2021-10-04
Added
	Add table foot option for sortable table.

[0.8.1] - 2021-08-11
Changed
	Loosen version requirement for Flop.

[0.8.0] - 2021-08-11
Added
	New options event and target for the pagination and sortable table
component, which allow to emit pagination and sorting events in LiveView
without patching the URL.

Changed
	Use HEEx templates for both the pagination and the sortable table component.
Refer to the Readme for usage examples.
	Require live_view ~> 0.16.0.
	Support safe HTML tuples in unsortable table headers.
	Improve documentation with examples for LiveView, HEEx templates and EEx
templates.

[0.7.0] - 2021-06-13
Added
	Add wrapper around sortable table header link and order direction indicator.
	Add option current_link_attrs to pagination builder.
	Add options thead_tr_attrs, thead_th_attrs, tbody_tr_attrs and
tbody_td_attrs to table generator.
	Add option no_results_content to table generator, which sets the content
that is going to be displayed instead of the table if the item list is empty.
A default option is applied, so make sure to set the option and/or remove your
own no results messages from your templates when making the upgrade.

Changed
	The table options table_class, th_wrapper_class, symbol_class and
container_class were replaced in favour of table_attrs,
th_wrapper_attrs, symbol_attrs and container_attrs for more flexibility
and consistency with the pagination generator. To update, rename the options
in your code and wrap the values in keyword lists with a class key
(e.g. container_class: "table-container" =>
container_attrs: [class: "table-container"]).
	The pagination_link_attrs is not applied to current links anymore. Use
current_link_attrs to set the attributes for the current link.
	Omit page=1 and offset=0 when building query parameters.
	Omit default values for order and limit/page size parameters when building
query parameters.
	Requires Flop ~> 0.11.0.

Fixed
	Order direction indicator was wrapped twice.
	A Flop struct with an offset was resulting in invalid pagination links.

[0.6.1] - 2021-05-05
Fixed
	Pagination helper generated invalid links when using default_limit option.

[0.6.0] - 2021-05-04
Added
	Add Flop.Phoenix.table/1 for rendering sortable tables.
	Add function Flop.Phoenix.to_query/1, which converts a Flop struct into
a keyword list for query parameters.
	Add function Flop.Phoenix.build_path/3, which applies Flop parameters to a
Phoenix path helper function.

Removed
	Remove Flop.Phoenix.Live.PaginationComponent in favor of making
Flop.Phoenix.pagination/4 work in both .eex and .leex templates.

[0.5.1] - 2021-04-14
Fixed
	Merge pagination query parameters into existing query parameters, if present.

[0.5.0] - 2021-03-23
Changed
	Rename FlopPhoenix to Flop.Phoenix.
	Add Flop.Phoenix.Live.PaginationComponent for use with Phoenix.LiveView.
	Change pagination to always display links to the first and last page.

[0.4.0] - 2020-09-04
Changed
	Allow usage with newer versions of Flop.

[0.3.0] - 2020-06-17
Added
	New option to hide the number of page links.
	New option to limit the number of page links.

Changed
	Add order and filter parameters to pagination links.

[0.2.0] - 2020-06-15
Added
	Improve documentation.

Changed
	previous_link/3, next_link/3 and page_links/3 are private functions now.

[0.1.0] - 2020-06-15
Initial release


  

    CSS Styles

Pagination
The pagination component doesn't set any default styles. While you can set class
attributes for every element, it is entirely possible to style the component
with just a base class.
<Flop.Phoenix.pagination class="pagination" ...>
  <%!-- ... --%>
</Flop.Phoenix.pagination>
The class above will be set on the <nav> element that wraps the pagination
controls. With this in place, we can apply some basic styles. For readability,
we're going to use nested CSS rules in this example.
.pagination {
  display: flex;
  gap: 0.25rem;
  align-items: center;
  justify-content: space-between;

  > ul {
    display: flex;
    flex: 1 1 auto;
    flex-wrap: wrap;
    gap: 0.25rem;
    align-items: center;
    justify-content: flex-end;
    padding: 0;
    margin: 0;
    list-style-type: none;

    li {
      flex: 0 1 auto;
      padding: 0;
      margin: 0;
    }
  }

  > a:first-child,
  > button:first-child {
    flex: none;
  }

  > a:last-of-type,
  > button:last-of-type {
    flex: none;
  }

  a,
  button,
  li > span {
    padding: 0.25rem 0.5rem;
    margin: 0;
    border-radius: 4px;
  }

  li > span {
    color: #00749d;
  }

  a,
  button {
    color: #fff;
    white-space: nowrap;
    background: #00749d;
    border: none;

    &:hover,
    &:focus {
      background-color: #003c54;
    }

    &[aria-current="page"] {
      background-color: #001d2a;
    }
  }

  [disabled],
  [aria-disabled="true"] {
    pointer-events: none;
    opacity: 0.7;
  }
}
Let's break this down.
.pagination {
  display: flex;
  gap: 0.25rem;
  align-items: center;
  justify-content: space-between;

  // ...
}
This selects the <nav> element by using the base class we passed to the
component as an attribute.
> ul {
  display: flex;
  flex: 1 1 auto;
  flex-wrap: wrap;
  gap: 0.25rem;
  align-items: center;
  justify-content: flex-end;
  padding: 0;
  margin: 0;
  list-style-type: none;

  li {
    flex: 0 1 auto;
    padding: 0;
    margin: 0;
  }
}
The numbered page links are rendered within an unordered list, which is a
direct descendant of the navigation element.
> a:first-child,
> button:first-child {
  flex: none;
}

> a:last-of-type,
> button:last-of-type {
  flex: none;
}
These are styles for the previous link or button (first child) and the next
link or button (last of type). Depending on whether the path attribute is
set on the component, Flop.Phoenix will either use <a> elements with an href
or <button> elements to adhere to proper semantics. We need to account for
this in our styles.
a,
button,
li > span {
  padding: 0.25rem 0.5rem;
  margin: 0;
  border-radius: 4px;
}

li > span {
  color: #00749d;
}
Here we set some basic styles for the links/buttons (both previous/next and
numbered page links). The li > span selector is used to style the ellipsis,
which uses a classless span by default.
a,
button {
  color: #fff;
  white-space: nowrap;
  background: #00749d;
  border: none;

  &:hover,
  &:focus {
    background-color: #003c54;
  }

  &[aria-current="page"] {
    background-color: #001d2a;
  }
}
Here we set some more styles on the links/buttons, without the ellipsis span.
The pagination component sets the aria-current attribute on the page number
link/button to the current page. We can bind our styles to this attributes
without needing a class.
[disabled],
[aria-disabled="true"] {
  pointer-events: none;
  opacity: 0.7;
}
Finally, we define styles for disabled previous/next links and buttons. If
buttons are used, the disabled attribute is set. Since the disabled
attribute cannot be used with links, the aria-disabled attribute is set on
<a> elements.
Changing the order
The pagination component will always render the same markup: Individual
previous/next links/buttons first, and then a list with page number
links/buttons. To visually change the order while maintaining the same markup,
you can use the order attribute.
To put the page number links/buttons first and the previous/next links/buttons
last:
> a:first-child,
> button:first-child {
  order: 2;
}

> a:last-of-type,
> button:last-of-type {
  order: 3;
}

> ul {
  order: 1;
}
To put the page number links/buttons in the middle between the previous
link/button and the next link/button:
> a:first-child,
> button:first-child {
  order: 1;
}

> a:last-of-type,
> button:last-of-type {
  order: 3;
}

> ul {
  order: 2;
}
Adjust the remaining flex properties as needed.


  

    Load More and Infinite Scroll

You can use Flop's cursor pagination to implement load-more buttons and infinite
scroll.
Let's start with a basic "Load More" button.
Load More Button
We'll define a basic schema:
defmodule Petshop.Pets.Pet do
  use Ecto.Schema
  import Ecto.Changeset

  @derive {Flop.Schema,
           filterable: [:name],
           sortable: [:id],
           default_order: %{order_by: [:id], order_directions: [:desc]},
           default_limit: 50}

  schema "pets" do
    field :name, :string

    timestamps(type: :utc_datetime)
  end
end
We derive the Flop.Schema protocol, make the id column sortable, set the
default order to id DESC, and set the default limit to 50.
Then we define a list function:
def list_pets(params) do
  Flop.validate_and_run!(Pet, params,
    for: Pet,
    default_pagination_type: :first,
    pagination_types: [:first],
    replace_invalid_params: true,
    filtering: false,
    ordering: false
  )
end
We set the default pagination type to :first, so that we get a response with
cursors even if no pagination parameters were passed. We chose :first because
the default order in the schema already sorts by ID descending. We also enable
replace_invalid_params, so that invalid parameters are silently ignored, and
we disable ordering and filtering via parameters.
In the handle_params function of our LiveView, we simply call the list
function, assign the meta struct, and stream the pets.
@impl true
def handle_params(params, _url, socket) do
  {pets, meta} = Pets.list_pets(params)
  {:noreply, socket |> stream(:pets, pets) |> assign(:meta, meta)}
end
In our HEEx template, we're going to use a basic Flop table to render the
pets (but you could use any other component here), and we render the
"Load More" link. We only render the link after checking whether there is
indeed a next page.
<Flop.Phoenix.table
  id="pets"
  items={@streams.pets}
  meta={@meta}
  on_sort={JS.push("sort-table")}
  row_click={fn {_id, pet} -> JS.navigate(~p"/pets/#{pet}") end}
>
  <:col :let={{_id, pet}} label="Name">{pet.name}</:col>
  <:action :let={{_id, pet}}>
    <div class="sr-only">
      <.link navigate={~p"/pets/#{pet}"}>Show</.link>
    </div>
  </:action>
</Flop.Phoenix.table>

<p :if={@meta.has_next_page?}>
  <.link patch={~p"/pets?after=#{@meta.end_cursor}"}>Load More</.link>
</p>
The "Load More" link just attaches the end cursor as after parameter to the
base URL. A click then triggers the handle_params function, which passes the
parameter on to the list function. And since the end cursor is in the URL, the
user can reload the page or the LiveView can be re-mounted without the user
losing their position in the list.
And that's it - you should now have a working "Load More" button.
Infinite Scroll
Now let's add infinite scroll to the view. First, we'll add a new hook to our
assets/app.js:
let Hooks = {};

Hooks.InfiniteScroll = {
  mounted() {
    observer = new IntersectionObserver(
      (entries) => {
        entries.forEach((entry) => {
          if (entry.isIntersecting) {
            this.pushEvent("load-more", {});
          }
        });
      },
      {
        root: null,
        rootMargin: "0px",
        threshold: 1.0,
      },
    );

    const anchorId = this.el.dataset.anchorId;
    const anchor = document.getElementById(anchorId);

    if (anchor) {
      observer.observe(anchor);
    } else {
      console.error(`Anchor element not found: ${anchorId}`);
    }
  },
};
We're using an IntersectionObserver to observe an anchor that is referenced with
a data attribute (data-anchor-id) on the hook element.
We then ensure to pass the hook when we create the LiveSocket:
let liveSocket = new LiveSocket("/live", Socket, {
  longPollFallbackMs: 2500,
  params: { _csrf_token: csrfToken },
  hooks: Hooks,
});
And that's it on the JavaScript side. In our HEEx template, we wrap our table
and "Load More" link into a div, to which we add the phx-hook attribute.
<div
  id="main-content"
  phx-hook="InfiniteScroll"
  data-anchor-id="load-more-anchor"
>
  <Flop.Phoenix.table ...>
    <%!-- etc. --%>
  </Flop.Phoenix.table>

  <p :if={@meta.has_next_page?} id="load-more-anchor">
    <.link patch={~p"/pets?after=#{@meta.end_cursor}"}>Load More</.link>
  </p>
</div>
Note that the paragraph that contains the "Load More" link acts as an anchor: We
gave it an ID and referenced it in the data-anchor-id attribute of the div.
We need one final change: The hook doesn't patch the URL directly, but pushes a
load-more event to the server. If we wanted to patch the URL directly with the
hook, we'd need to tell it the base URL and the end cursor. While possible, it's
probably easier to handle this on the server-side.
To handle the event, we need to add a handle_event function to our LiveView.
It checks whether there is indeed an end cursor and patches the URL just like
a click on the "Load More" link would do.
def handle_event("load-more", _, socket) do
  if end_cursor = socket.assigns.meta.end_cursor do
    {:noreply, push_patch(socket, to: ~p"/pets?after=#{end_cursor}")}
  else
    {:noreply, socket}
  end
end
Since the paragraph with the "Load More" link is only shown if there are more
items to show, there won't be any unnecessary events when we reach the end of
the list. The visible and focusable "Load More" button serves as a fallback for
users who rely on keyboard navigation or assistive technologies.
Final Thoughts
This guide only described a basic implementation of infinite scroll using Flop.
Some details have been omitted, for example:
	If you reload the page with the after parameter set, the list will start
where you left off, but you cannot view the previous items anymore. You might
want to add another "Load More" button at the top of the list.
	The container should have the
"feed" role, which also has recommended keyboard interactions.
	You might want to support the browser's back button.
	You might want to give users the option to disable infinite scrolling.

Apart from the technical side, infinite scroll has a lot of accessibility and
usability issues. It can interfere with assistive technologies, make it
harder to navigate or return to a specific spot, prevent a sense of
progress or place, and reduce user agency.
Before you use infinite scroll, consider whether it's truly the best fit for
your users. In many cases, conventional pagination or the use of a "Load More"
button without infinite scroll can provide a more user-friendly experience.
Resources
	Infinite Scrolling & Role=Feed Accessibility Issues
	Infinite Scrolling: When to Use It, When to Avoid It
	Humane by Design Principles: Finite
	Infinite Scroll & Accessibility! Is It Any Good?
	OK ARIA! Role=feed is here & it’s not ready for prime time
	So You Think You’ve Built a Good Infinite Scroll



  

    Page Size Control

To allow users to change the page size, you can create a component similar to
this:
attr :current_size, :integer, required: true

def page_size_links(assigns) do
  ~H"""
  <ul class="page-size-links">
    <li
      :for={page_size <- [10, 20, 40, 60]}
      class={page_size == @current_size && "is-active"}
    >
      <.link phx-click="set-page-size" phx-value-size={page_size}>
        {page_size}
      </.link>
    </li>
  </ul>
  """
end
Render it with:
<.page_size_links current_size={@meta.page_size} />
And handle the event:
  def handle_event("set-page-size", %{"size" => page_size}, socket) do
    flop = %{socket.assigns.meta.flop | page_size: page_size, limit: nil}
    path = Flop.Phoenix.build_path(~p"/pets", flop)

    {:noreply, push_patch(socket, to: path)}
  end


  

    
Flop.Phoenix 
    



      
Phoenix components for pagination, sortable tables and filter forms with
Flop.
Introduction
Please refer to the Readme for an introduction.
Customization
To customize the components, it is recommended to define wrapper components in
your CoreComponents module that pass attributes that are constant for your
application and add additional markup as necessary.
For example, to customize the pagination component, define your own
pagination component:
defmodule MyAppWeb.CoreComponents do
  use Phoenix.Component

  attr :meta, Flop.Meta, required: true
  attr :path, :any, default: nil
  attr :on_paginate, JS, default: nil
  attr :target, :string, default: nil

  attr :aria_label, :string,
    default: "Pagination",
    doc: """
    Aria label for the `<nav>` element. The value should be localized. In
    languages with latin characters, the first letter should be capitalized.
    If multiple pagination components are rendered on the same page, each one
    should have a distinct aria label.
    """

  def pagination(assigns) do
    ~H"""
    <Flop.Phoenix.pagination
      class="pagination"
      meta={@meta}
      path={@path}
      on_paginate={@on_paginate}
      target={@target}
      aria-label={@aria_label}
      page_link_aria_label_fun={&"#{&1}ページ目へ"}
    >
      <:previous attrs={[class: "previous"]}>
        <i class="fas fa-chevron-left"/>
      </:previous>
      <:next attrs={[class: "next"]}>
        <i class="fas fa-chevron-right"/>
      </:next>
      <:ellipsis>
        <span class="ellipsis">‥</span>
      </:ellipsis>
    </Flop.Phoenix.pagination>
    """
  end
end
Refer to the documentation of Flop.Phoenix.pagination/1 for available
attributes and slots on the pagination component and to table_option/0 for
a list of available options and defaults for the table component.
Using links
If the path attribute is set on the pagination and table component,
pagination and sorting is handled via query parameters. You will need to
handle those parameters in the Phoenix.LiveView.handle_params/3 callback
of your LiveView module.
def handle_params(params, _, socket) do
  {pets, meta} = Pets.list_pets(params)
  {:noreply, assign(socket, pets: pets, meta: meta)}
end
Using JS commands
You can pass a Phoenix.LiveView.JS command as on_paginate and on_sort
attributes.
If used with the path attribute, the URL will be patched and the given
JS command will be executed. This can be used to scroll to the top after a
pagination or sorting event, for example.
If used without the path attribute, you will need to include a push
command to trigger an event when a pagination or sort link is clicked.
You can set a different target with the target attribute, which will be used
as phx-target.
<Flop.Phoenix.table
  items={@items}
  meta={@meta}
  on_sort={JS.push("sort-pets")}
  target={@myself}
/>

<Flop.Phoenix.pagination
  meta={@meta}
  on_paginate={JS.push("paginate-pets")}
  target={@myself}
/>
You will need to handle the event in the Phoenix.LiveView.handle_event/3
or Phoenix.LiveComponent.handle_event/3 callback of your
LiveView or LiveComponent module.
# for page-based pagination
def handle_event("paginate-pets", %{"page" => page}, socket) do
  flop = Flop.set_page(socket.assigns.meta.flop, page)
  {pets, meta} = Pets.list_pets(flop)
  {:noreply, assign(socket, pets: pets, meta: meta)}
end

# for cursor-based pagination
def handle_event("paginate-pets", %{"to" => direction}, socket) do
  flop =
    case direction do
      :previous -> Flop.to_previous_cursor(socket.assigns.meta)
      :next -> Flop.to_next_cursor(socket.assigns.meta)
    end

  {pets, meta} = Pets.list_pets(flop)
  {:noreply, assign(socket, pets: pets, meta: meta)}
end

def handle_event("sort-pets", %{"order" => order}, socket) do
  flop = Flop.push_order(socket.assigns.meta.flop, order)
  {pets, meta} = Pets.list_pets(flop)
  {:noreply, assign(socket, pets: pets, meta: meta)}
end

      


      
        Summary


  
    Components
  


    
      
        filter_fields(assigns)

      


        Renders all inputs for a filter form including the hidden inputs.



    


    
      
        hidden_inputs_for_filter(assigns)

      


        Renders hidden inputs for the given form.



    


    
      
        pagination(assigns)

      


        Renders pagination links for the given Flop.Meta struct.



    


    
      
        pagination_for(assigns)

      


        This component is a pagination builder.



    


    
      
        table(assigns)

      


        Generates a table with sortable columns.



    





  
    Types
  


    
      
        page_link_option()

      


        Defines how many page links to render.



    


    
      
        table_option()

      


        Defines the available options for Flop.Phoenix.table/1.



    





  
    Functions
  


    
      
        build_path(path, meta_or_flop_or_params, opts \\ [])

      


        Builds a path that includes query parameters for the given Flop struct
using the referenced Phoenix path helper function.



    


    
      
        page_link_aria_label(n)

      


        Returns an aria label for a link to the given page number.



    


    
      
        page_link_range(max_pages, current_page, total_pages)

      


        Returns the range of page links to be rendered.



    


    
      
        to_query(flop, opts \\ [])

      


        Converts a Flop struct into a keyword list that can be used as a query with
Phoenix verified routes or route helper functions.



    





      


      
        Components

        


  
    
      
    
    
      filter_fields(assigns)


        (since 0.12.0)


        
          
        

    

  


  

      

          @spec filter_fields(map()) :: Phoenix.LiveView.Rendered.t()


      


Renders all inputs for a filter form including the hidden inputs.
Example
def filter_form(%{meta: meta} = assigns) do
  assigns = assign(assigns, :form, Phoenix.Component.to_form(meta))

  ~H"""
  <.form for={@form}>
    <.filter_fields :let={i} form={@form} fields={[:email, :name]}>
      <.input
        field={i.field}
        label={i.label}
        type={i.type}
        {i.rest}
      />
    </.filter_fields>
  </.form>
  """
end
This assumes that you have defined an input component that renders a form
input including the label.
These options are passed to the inner block via :let:
	The field is a Phoenix.HTML.FormField struct.
	The type is the input type as a string (e.g. "text", "number"). The
type is derived from the type of the field being filtered on, but it can
be overridden in the field options.
	rest contains any additional field options passed.

Field configuration
The fields can be passed as atoms or keywords with additional options.
fields={[:name, :email]}
Or
fields={[
  name: [label: gettext("Name")],
  email: [
    label: gettext("Email"),
    op: :ilike_and,
    type: "email"
  ],
  age: [
    label: gettext("Age"),
    type: "select",
    prompt: "",
    options: [
      {gettext("young"), :young},
      {gettext("old"), :old)}
    ]
  ]
]}
Available options:
	label - Defaults to the humanized field name.
	op - Defaults to :==.
	type - Defaults to an input type depending on the Ecto type of the filter
field.

Any additional options will be passed to the input component (e.g. HTML
classes or a list of options).
Attributes
	form (Phoenix.HTML.Form) (required)

	fields (:list) - The list of fields and field options. Note that inputs will not be rendered
for fields that are not marked as filterable in the schema
(see Flop.Schema).
If dynamic is set to false, only fields in this list are rendered. If
dynamic is set to true, only fields for filters present in the given
Flop.Meta struct are rendered, and the fields are rendered even if they
are not passed in the fields list. In the latter case, fields is
optional, but you can still pass label and input configuration this way.
Note that in a dynamic form, it is not possible to configure a single field
multiple times.
Defaults to [].

	dynamic (:boolean) - If true, fields are only rendered for filters that are present in the
Flop.Meta struct passed to the form. You can use this for rendering filter
forms that allow the user to add and remove filters dynamically. The
fields assign is only used for looking up the options in that case.
Defaults to false.


Slots
	inner_block - The necessary options for rendering a label and an input are passed to the
inner block, which allows you to render the fields with your existing
components.<.filter_fields :let={i} form={@form} fields={[:email, :name]}>
  <.input
    field={i.field}
    label={i.label}
    type={i.type}
    {i.rest}
  />
</.filter_fields>
The options passed to the inner block are:	field - A Phoenix.HTML.FormField struct.
	type - The input type as a string.
	label - The label text as a string.
	rest - Any additional options passed in the field options.




  



  
    
      
    
    
      hidden_inputs_for_filter(assigns)


        (since 0.16.0)


        
          
        

    

  


  

Renders hidden inputs for the given form.
You can use this for convenience if you have a complex form layout that cannot
be accomplished with Flop.Phoenix.filter_fields/1. Put it as a direct child
of the form component to render the hidden inputs for pagination and order
parameters. Then use Phoenix.Component.inputs_for/1 to render a single
filter field, and place this component within the anonymous function to render
the hidden inputs for the filter field and operator.
Since the filters are represented as an array in the params, make sure to
add the offset option so that the Flop.Meta can be properly mapped back to
your input fields. For every call to inputs_for always add the length of all
previous calls to inputs_for as offset. Note that the example below uses
the old Phoenix.Component.inputs_for/1 function.
Also don't forget to set the skip_persistent_id attribute
to prevent LiveView from overriding the IDs and causing duplicate DOM IDs.
<.form :let={f} for={@meta}>
  <.hidden_inputs_for_filter form={@form} />

  <div class="field-group">
    <div class="field">
      <.inputs_for
        :let={ff}
        field={f[:filters]}
        options={[fields: [:name]]}
        skip_persistent_id
      >
        <.hidden_inputs_for_filter form={ff} />
        <.input label="Name" type="text" field={ff[:value]} />
      </.inputs_for>
    </div>
    <div class="field">
      <.inputs_for
        :let={ff}
        field={f[:filters]}
        options={[fields: [email: [op: :ilike]], offset: 1]}
        skip_persistent_id
      >
        <.hidden_inputs_for_filter form={ff} />
        <.input label="E-mail" type="email" field={ff[:value]} />
      </.inputs_for>
    </div>
  </div>
</.form>
Attributes
	form (Phoenix.HTML.Form) (required)

          @spec pagination(map()) :: Phoenix.LiveView.Rendered.t()


      


Renders pagination links for the given Flop.Meta struct.
This component can render both page-based or cursor-based pagination links.
Which one is used depends on the pagination type that was used to make the
Flop query.
Examples
With a verified route:
<Flop.Phoenix.pagination
  meta={@meta}
  path={~p"/pets"}
/>
With an event:
<Flop.Phoenix.pagination
  meta={@meta}
  on_paginate={JS.push("paginate")}
/>
With a route helper:
<Flop.Phoenix.pagination
  meta={@meta}
  path={{Routes, :pet_path, [@socket, :index]}}
/>
With all attributes and slots:
<Flop.Phoenix.pagination
  meta={@meta}
  path={~p"/pets"}
  on_paginate={JS.dispatch("my_app:scroll_to", to: "#pet-table")}
  target={@myself}
  class="pagination"
  page_link_aria_label_fun={&"#{&1}ページ目へ"}
  page_links={5}
  page_list_attrs={[class: "pagination-list"]}
  page_list_item_attrs={[class: "pagination-item"]}
  page_link_attrs={[class: "pagination-link"]}
  current_page_link_attrs={[class: "pagination-link is-current"]}
  disabled_link_attrs={[class: "is-disabled"]}
  reverse
>
  <:previous attrs={[class: "pagination-previous"]}>
    Previous
  </:previous>
  <:next attrs={[class: "pagination-next"]}>
    Next
  </:next>
  <:ellipsis>
    <span class="pagination-ellipsis" aria-hidden="true">&hellip;</span>
  </:ellipsis>
</Flop.Phoenix.pagination>
Attributes
	meta (Flop.Meta) (required) - The meta information of the query as returned by the Flop query functions.

	path (:any) - If set, the current view is patched with updated query parameters when a
pagination link is clicked. In case the on_paginate attribute is set as
well, the URL is patched and the given command is executed.
The value must be either a URI string (Phoenix verified route), an MFA or FA
tuple (Phoenix route helper), or a 1-ary path builder function. See
Flop.Phoenix.build_path/3 for details.
Defaults to nil.

	on_paginate (Phoenix.LiveView.JS) - A Phoenix.LiveView.JS command that is triggered when a pagination link is
clicked.
If used without the path attribute, you should include a push operation
to handle the event with the handle_event callback.
<.pagination
  meta={@meta}
  on_paginate={
    JS.dispatch("my_app:scroll_to", to: "#pet-table") |> JS.push("paginate")
  }
/>
If used with the path attribute, the URL is patched and the given
JS command is executed.
<.pagination
  meta={@meta}
  path={~p"/pets"}
  on_paginate={JS.dispatch("my_app:scroll_to", to: "#pet-table")}
/>
With the above attributes in place, you can add the following JavaScript to
your application to scroll to the top of your table whenever a pagination
link is clicked:
window.addEventListener("my_app:scroll_to", (e) => {
  e.target.scrollIntoView();
});
You can use CSS to scroll to the new position smoothly.
html {
  scroll-behavior: smooth;
}
Defaults to nil.

	target (:string) - Sets the phx-target attribute for the pagination links. Defaults to nil.

	page_link_aria_label_fun ({:fun, 1}) - Function that returns an aria label for the page link or button to the
given page number.
The returned label should be localized and start with a capital letter.
Defaults to &Flop.Phoenix.page_link_aria_label/1.

	page_links (:any) - Defines how many page links to render.
	:all - Renders all page links.
	:none - Does not render any page links.
	Integer - Renders up to the specified number of page links in addition to
the first and last page.

This attribute is only used for page-based pagination.
Defaults to 5.

	reverse (:boolean) - By default, the next link moves forward with the :after parameter set to
the end cursor, and the previous link moves backward with the :before
parameter set to the start cursor. If reverse is set to true, the
destinations of the links are switched.
This attribute is only for cursor-based pagination.
Defaults to false.

	page_list_attrs (:list) - Attributes to be added to the <ul> that contains the page links. Defaults to [].

	page_list_item_attrs (:list) - Attributes to be added to the <li> elements that contain the page links. Defaults to [].

	page_link_attrs (:list) - Attributes to be added to the page links or buttons.
These attributes are not applied to previous links, next links, or the
current page link.
Defaults to [].

	current_page_link_attrs (:list) - Attributes to be added to the current page link or button.
Note that the aria-current attribute is automatically set.
It is recommended to define CSS styles using the [aria-current="page"]
selector instead of using a class.
Defaults to [].

	disabled_link_attrs (:list) - Attributes to be added to disabled previous/next links or buttons.
If a class is set, it is merged with the class set on the previous/next
slot.
Note that the disabled attribute is automatically set for buttons and the
aria-disabled="true" attribute is automatically set for links.
It is recommended to define CSS styles using the
[disabled], [aria-disabled="true"] selector instead of using a class.
Defaults to [].

	Global attributes are accepted. The attributes are added to the outer <nav> element.
The aria-label defaults to "Pagination". If your application is
localized, the label should be translated to the user locale. In languages
with latin characters, the first letter should be capitalized. If multiple
pagination components are rendered on the same page, each one should have
a distinct aria label.


Slots
	previous - The content of the pagination link or button to the previous page.
If the slot is not used, the text "Previous" is rendered.
Accepts attributes:
	attrs (:list) - Any additional attributes to add to the link or button.
Defaults to [aria: [label: "Go to previous page"]].



	next - The content of the pagination link or button to the next page.
If the slot is not used, the text "Next" is rendered.
Accepts attributes:
	attrs (:list) - Any additional attributes to add to the link or button.
Defaults to [aria: [label: "Go to next page"]].



	ellipsis - The content of the <li> element that usually shows an ellipsis and is
rendered toward the beginning and/or end of the page links if there are
more pages than the configured limit.
If the slot is not used, a default element is used:
<span aria-hidden="true">&hellip;</span>



  



  
    
      
    
    
      pagination_for(assigns)


        (since 0.24.0)


        
          
        

    

  


  

      

          @spec pagination_for(map()) :: Phoenix.LiveView.Rendered.t()


      


This component is a pagination builder.
It does not render anything by itself. Instead, it prepares all the necessary
information needed to render a pagination component and passes it to the
inner block.
For an example implementation, see pagination/1.
Example
<.pagination_for
  :let={p}
  meta={@meta}
  page_links={4}
  path={~p"/birds"}
>
  <%!-- put together your component here --%>
</.pagination_for>
The variable passed to the inner block is a Flop.Phoenix.Pagination struct.
Attributes
	meta (Flop.Meta) (required) - The meta information of the query as returned by the Flop query functions.

	path (:any) - If set, a function that takes a page number and returns a link with
pagination, filter, and sort parameters based on the given path is passed
as path_fun to the inner block.
The value must be either a URI string (Phoenix verified route), an MFA or FA
tuple (Phoenix route helper), or a 1-ary path builder function. See
Flop.Phoenix.build_path/3 for details.
Defaults to nil.

	page_links (:any) - Defines how many page links to render.
	:all - Renders all page links.
	:none - Does not render any page links.
	Integer - Renders up to the specified number of page links in addition to
the first and last page.

A page_range_start and page_range_end attribute are passed to the
inner block based on this option. If this attribute is set to :none, both
of those values will be nil.
Defaults to 5.

	reverse (:boolean) - By default, the next link moves forward with the :after parameter set to
the end cursor, and the previous link moves backward with the :before
parameter set to the start cursor. If reverse is set to true, the
destinations of the links are switched.
Defaults to false.


Slots
	inner_block (required)


  



  
    
      
    
    
      table(assigns)


        (since 0.6.0)


        
          
        

    

  


  

      

          @spec table(map()) :: Phoenix.LiveView.Rendered.t()


      


Generates a table with sortable columns.
Example
<Flop.Phoenix.table items={@pets} meta={@meta} path={~p"/pets"}>
  <:col :let={pet} label="Name" field={:name}>{pet.name}</:col>
  <:col :let={pet} label="Age" field={:age}>{pet.age}</:col>
</Flop.Phoenix.table>
Flop.Schema
If you pass the for option when making the query with Flop, Flop Phoenix can
determine which table columns are sortable. It also hides the order and
page_size parameters if they match the default values defined with
Flop.Schema.
Attributes
	id (:string) - ID used on the table. If not set, an ID is chosen based on the schema
module derived from the Flop.Meta struct.
The ID is necessary in case the table is fed with a LiveView stream.

	items (:list) (required) - The list of items to be displayed in rows. This is the result list returned
by the query.

	meta (Flop.Meta) (required) - The Flop.Meta struct returned by the query function.

	path (:any) - If set, the current view is patched with updated query parameters when a
header link for sorting is clicked. In case the on_sort attribute is
set as well, the URL is patched and the given JS command is executed.
The value must be either a URI string (Phoenix verified route), an MFA or FA
tuple (Phoenix route helper), or a 1-ary path builder function. See
Flop.Phoenix.build_path/3 for details.
Defaults to nil.

	on_sort (Phoenix.LiveView.JS) - A Phoenix.LiveView.JS command that is triggered when a header link for
sorting is clicked.
If used without the path attribute, you should include a push operation
to handle the event with the handle_event callback.
<.table
  items={@items}
  meta={@meta}
  on_sort={
    JS.dispatch("my_app:scroll_to", to: "#pet-table") |> JS.push("sort")
  }
/>
If used with the path attribute, the URL is patched and the given
JS command is executed.
<.table
  meta={@meta}
  path={~p"/pets"}
  on_sort={JS.dispatch("my_app:scroll_to", to: "#pet-table")}
/>
Defaults to nil.

	target (:string) - Sets the phx-target attribute for the header links. Defaults to nil.

	caption (:string) - Content for the <caption> element. Defaults to nil.

	opts (:list) - Keyword list with additional options (see Flop.Phoenix.table_option/0).
Note that the options passed to the function are deep merged into the
default options. Since these options will likely be the same for all the
tables in a project, it is recommended to define them once in a function or
set them in a wrapper function as described in the Customization section
of the module documentation.
Defaults to [].

	row_id (:any) - Overrides the default function that retrieves the row ID from a stream item. Defaults to nil.

	row_click (:any) - Sets the phx-click function attribute for each row td. Expects to be a
function that receives a row item as an argument. This does not add the
phx-click attribute to the action slot.
Example:
row_click={&JS.navigate(~p"/users/#{&1}")}
Defaults to nil.

	row_item (:any) - This function is called on the row item before it is passed to the :col
and :action slots.
Defaults to &Function.identity/1.


Slots
	col (required) - For each column to render, add one <:col> element.
<:col :let={pet} label="Name" field={:name} col_style="width: 20%;">
  {pet.name}
</:col>
Any additional assigns will be added as attributes to the <td> elements.
Accepts attributes:
	label (:any) - The content for the header column.

	field (:atom) - The field name for sorting. If set and the field is configured as sortable
in the schema, the column header will be clickable, allowing the user to
sort by that column. If the field is not marked as sortable or if the
field attribute is omitted or set to nil or false, the column header
will not be clickable.

	directions (:any) - An optional 2-element tuple used for custom ascending and descending sort
behavior for the column, i.e. {:asc_nulls_last, :desc_nulls_first}

	col_style (:string) - If set, a <colgroup> element is rendered and the value of the
col_style assign is set as style attribute for the <col> element of
the respective column. You can set the width, background, border,
and visibility of a column this way.

	col_class (:string) - If set, a <colgroup> element is rendered and the value of the
col_class assign is set as class attribute for the <col> element of
the respective column. You can set the width, background, border,
and visibility of a column this way.

	thead_th_attrs (:list) - Additional attributes to pass to the <th> element as a static keyword
list. Note that these attributes will override any conflicting
thead_th_attrs that are set at the table level.

	th_wrapper_attrs (:list) - Additional attributes for the <span> element that wraps the
header link and the order direction symbol. Note that these attributes
will override any conflicting th_wrapper_attrs that are set at the table
level.

	tbody_td_attrs (:any) - Additional attributes to pass to the <td> element. May be provided as a
static keyword list, or as a 1-arity function to dynamically generate the
list using row data. Note that these attributes will override any
conflicting tbody_td_attrs that are set at the table level.



	action - The slot for showing user actions in the last table column. These columns
do not receive the row_click attribute.
<:action :let={user}>
  <.link navigate={~p"/users/#{user}"}>Show</.link>
</:action>
Accepts attributes:
	label (:string) - The content for the header column.

	col_style (:string) - If set, a <colgroup> element is rendered and the value of the
col_style assign is set as style attribute for the <col> element of
the respective column. You can set the width, background, border,
and visibility of a column this way.

	col_class (:string) - If set, a <colgroup> element is rendered and the value of the
col_class assign is set as class attribute for the <col> element of
the respective column. You can set the width, background, border,
and visibility of a column this way.

	thead_th_attrs (:list) - Any additional attributes to pass to the <th> as a keyword list.

	tbody_td_attrs (:any) - Any additional attributes to pass to the <td>. Can be a keyword list or
a function that takes the current row item as an argument and returns a
keyword list.



	foot - You can optionally add a foot. The inner block will be rendered inside
a tfoot element.
<Flop.Phoenix.table>
  <:foot>
    <tr><td>Total: <span class="total">{@total}</span></td></tr>
  </:foot>
</Flop.Phoenix.table>

          @type page_link_option() :: :all | :none | pos_integer()


      


Defines how many page links to render.
	:all - Renders all page links.
	:none - Does not render any page links.
	Integer - Renders up to the specified number of page links in addition to
the first and last page.


  



  
    
      
    
    
      table_option()



        
          
        

    

  


  

      

          @type table_option() ::
  {:container, boolean()}
  | {:container_attrs, keyword()}
  | {:no_results_content, Phoenix.HTML.safe() | binary()}
  | {:symbol_asc, Phoenix.HTML.safe() | binary()}
  | {:symbol_attrs, keyword()}
  | {:symbol_desc, Phoenix.HTML.safe() | binary()}
  | {:symbol_unsorted, Phoenix.HTML.safe() | binary()}
  | {:table_attrs, keyword()}
  | {:tbody_attrs, keyword()}
  | {:thead_attrs, keyword()}
  | {:tbody_td_attrs, keyword()}
  | {:tbody_tr_attrs, keyword() | (any() -> keyword())}
  | {:th_wrapper_attrs, keyword()}
  | {:thead_th_attrs, keyword()}
  | {:thead_tr_attrs, keyword()}


      


Defines the available options for Flop.Phoenix.table/1.
	:container - Wraps the table in a <div> if true.
Default: false.
	:container_attrs - The attributes for the table container.
Default: [class: "table-container"].
	:no_results_content - Any content that should be rendered if there are no
results. Default: <p>No results.</p>.
	:table_attrs - The attributes for the <table> element.
Default: [].
	:th_wrapper_attrs - The attributes for the <span> element that wraps the
header link and the order direction symbol.
Default: [].
	:symbol_asc - The symbol that is used to indicate that the column is
sorted in ascending order.
Default: "▴".
	:symbol_attrs - The attributes for the <span> element that wraps the
order direction indicator in the header columns.
Default: [class: "order-direction"].
	:symbol_desc - The symbol that is used to indicate that the column is
sorted in ascending order.
Default: "▾".
	:symbol_unsorted - The symbol that is used to indicate that the column is
not sorted. Default: nil.
	:tbody_attrs: Attributes to be added to the <tbody> tag within the
<table>. Default: [].
	:tbody_td_attrs: Attributes to be added to each <td> tag within the
<tbody>. Default: [].
	:thead_attrs: Attributes to be added to the <thead> tag within the
<table>. Default: [].
	:tbody_tr_attrs: Attributes to be added to each <tr> tag within the
<tbody>. A function with arity of 1 may be passed to dynamically generate
the attrs based on row data.
Default: [].
	:thead_th_attrs: Attributes to be added to each <th> tag within the
<thead>. Default: [].
	:thead_tr_attrs: Attributes to be added to each <tr> tag within the
<thead>. Default: [].

      build_path(path, meta_or_flop_or_params, opts \\ [])


        (since 0.6.0)


        
          
        

    

  


  

      

          @spec build_path(
  String.t()
  | {module(), atom(), [any()]}
  | {function(), [any()]}
  | (keyword() -> String.t()),
  Flop.Meta.t() | Flop.t() | keyword(),
  keyword()
) :: String.t()


      


Builds a path that includes query parameters for the given Flop struct
using the referenced Phoenix path helper function.
The first argument can be either one of:
	an MFA tuple (module, function name as atom, arguments)
	a 2-tuple (function, arguments)
	a URL string, usually produced with a verified route (e.g. ~p"/some/path")
	a function that takes the Flop parameters as a keyword list as an argument

Default values for limit, page_size, order_by and order_directions are
omitted from the query parameters. To pick up the default parameters from a
schema module deriving Flop.Schema, you need to pass the :for option. To
pick up the default parameters from the backend module, you need to pass the
:backend option. If you pass a Flop.Meta struct as the second argument,
these options are retrieved from the struct automatically.
Date and Time Filters
When using filters on Date, DateTime, NaiveDateTime or Time fields,
you may need to implement the Phoenix.Param protocol for these structs.
See the documentation for to_query/2.
Examples
With a verified route
The examples below use plain URL strings without the p-sigil, so that the
doc tests work, but in your application, you can use verified routes or
anything else that produces a URL.
iex> flop = %Flop{page: 2, page_size: 10}
iex> path = build_path("/pets", flop)
iex> %URI{path: parsed_path, query: parsed_query} = URI.parse(path)
iex> {parsed_path, URI.decode_query(parsed_query)}
{"/pets", %{"page" => "2", "page_size" => "10"}}
The Flop query parameters will be merged into existing query parameters.
iex> flop = %Flop{page: 2, page_size: 10}
iex> path = build_path("/pets?species=dogs", flop)
iex> %URI{path: parsed_path, query: parsed_query} = URI.parse(path)
iex> {parsed_path, URI.decode_query(parsed_query)}
{"/pets", %{"page" => "2", "page_size" => "10", "species" => "dogs"}}
With an MFA tuple
iex> flop = %Flop{page: 2, page_size: 10}
iex> build_path(
...>   {Flop.PhoenixTest, :route_helper, [%Plug.Conn{}, :pets]},
...>   flop
...> )
"/pets?page_size=10&page=2"
With a function/arguments tuple
iex> pet_path = fn _conn, :index, query ->
...>   "/pets?" <> Plug.Conn.Query.encode(query)
...> end
iex> flop = %Flop{page: 2, page_size: 10}
iex> build_path({pet_path, [%Plug.Conn{}, :index]}, flop)
"/pets?page_size=10&page=2"
We're defining fake path helpers for the scope of the doctests. In a real
Phoenix application, you would pass something like
{Routes, :pet_path, args} or {&Routes.pet_path/3, args} as the
first argument.
Passing a Flop.Meta struct or a keyword list
You can also pass a Flop.Meta struct or a keyword list as the third
argument.
iex> pet_path = fn _conn, :index, query ->
...>   "/pets?" <> Plug.Conn.Query.encode(query)
...> end
iex> flop = %Flop{page: 2, page_size: 10}
iex> meta = %Flop.Meta{flop: flop}
iex> build_path({pet_path, [%Plug.Conn{}, :index]}, meta)
"/pets?page_size=10&page=2"
iex> query_params = to_query(flop)
iex> build_path({pet_path, [%Plug.Conn{}, :index]}, query_params)
"/pets?page_size=10&page=2"
Additional path parameters
If the path helper takes additional path parameters, just add them to the
second argument.
iex> user_pet_path = fn _conn, :index, id, query ->
...>   "/users/#{id}/pets?" <> Plug.Conn.Query.encode(query)
...> end
iex> flop = %Flop{page: 2, page_size: 10}
iex> build_path({user_pet_path, [%Plug.Conn{}, :index, 123]}, flop)
"/users/123/pets?page_size=10&page=2"
Additional query parameters
If the last path helper argument is a query parameter list, the Flop
parameters are merged into it.
iex> pet_url = fn _conn, :index, query ->
...>   "https://pets.flop/pets?" <> Plug.Conn.Query.encode(query)
...> end
iex> flop = %Flop{order_by: :name, order_directions: [:desc]}
iex> build_path({pet_url, [%Plug.Conn{}, :index, [user_id: 123]]}, flop)
"https://pets.flop/pets?user_id=123&order_directions[]=desc&order_by=name"
iex> build_path(
...>   {pet_url,
...>    [%Plug.Conn{}, :index, [category: "small", user_id: 123]]},
...>   flop
...> )
"https://pets.flop/pets?category=small&user_id=123&order_directions[]=desc&order_by=name"
Set page as path parameter
Finally, you can also pass a function that takes the Flop parameters as
a keyword list as an argument. Default values will not be included in the
parameters passed to the function. You can use this if you need to set some
of the parameters as path parameters instead of query parameters.
iex> flop = %Flop{page: 2, page_size: 10}
iex> build_path(
...>   fn params ->
...>     {page, params} = Keyword.pop(params, :page)
...>     query = Plug.Conn.Query.encode(params)
...>     if page, do: "/pets/page/#{page}?#{query}", else: "/pets?#{query}"
...>   end,
...>   flop
...> )
"/pets/page/2?page_size=10"
Note that in this example, the anonymous function just returns a string. With
Phoenix 1.7, you will be able to use verified routes.
build_path(
  fn params ->
    {page, query} = Keyword.pop(params, :page)
    if page, do: ~p"/pets/page/#{page}?#{query}", else: ~p"/pets?#{query}"
  end,
  flop
)
Note that the keyword list passed to the path builder function is built using
Plug.Conn.Query.encode/2, which means filters are formatted as map with
integer keys.
Set filter value as path parameter
If you need to set a filter value as a path parameter, you can use
Flop.Filter.pop/3.
iex> flop = %Flop{
...>   page: 5,
...>   order_by: [:published_at],
...>   filters: [
...>     %Flop.Filter{field: :category, op: :==, value: "announcements"}
...>   ]
...> }
iex> build_path(
...>   fn params ->
...>     {page, params} = Keyword.pop(params, :page)
...>     filters = Keyword.get(params, :filters, [])
...>     {category, filters} = Flop.Filter.pop(filters, :category)
...>     params = Keyword.put(params, :filters, filters)
...>     query = Plug.Conn.Query.encode(params)
...>
...>     case {page, category} do
...>       {nil, nil} -> "/articles?#{query}"
...>       {page, nil} -> "/articles/page/#{page}?#{query}"
...>       {nil, %{value: category}} -> "/articles/category/#{category}?#{query}"
...>       {page, %{value: category}} -> "/articles/category/#{category}/page/#{page}?#{query}"
...>     end
...>   end,
...>   flop
...> )
"/articles/category/announcements/page/5?order_by[]=published_at"

  



  
    
      
    
    
      page_link_aria_label(n)



        
          
        

    

  


  

      

          @spec page_link_aria_label(integer()) :: String.t()


      


Returns an aria label for a link to the given page number.
This is the default function used by pagination/1.
Example
iex> page_link_aria_label(5)
"Go to page 5"

  



  
    
      
    
    
      page_link_range(max_pages, current_page, total_pages)


        (since 0.24.0)


        
          
        

    

  


  

      

          @spec page_link_range(page_link_option(), pos_integer(), pos_integer()) ::
  {pos_integer() | nil, pos_integer() | nil}


      


Returns the range of page links to be rendered.
Usage
iex> page_link_range(:all, 4, 20)
{1, 20}

iex> page_link_range(:none, 4, 20)
{nil, nil}

iex> page_link_range(5, 4, 20)
{2, 6}

  



    

  
    
      
    
    
      to_query(flop, opts \\ [])


        (since 0.6.0)


        
          
        

    

  


  

Converts a Flop struct into a keyword list that can be used as a query with
Phoenix verified routes or route helper functions.
Default parameters
Default parameters for the limit and order parameters are omitted. The
defaults are determined by calling Flop.get_option/3.
	Pass the :for option to pick up the default values from a schema module
deriving Flop.Schema.
	Pass the :backend option to pick up the default values from your backend
configuration.
	If neither the schema module nor the backend module have default options
set, the function will fall back to the application environment.

Encoding queries
To encode the returned query as a string, you will need to use
Plug.Conn.Query.encode/1. URI.encode_query/1 does not support bracket
notation for arrays and maps.
Date and time filters
If you use the result of this function directly with
Phoenix.VerifiedRoutes.sigil_p/2 for verified routes or in a route helper
function, all cast filter values need to be able to be converted to a string
using the Phoenix.Param protocol.
This protocol is implemented by default for integers, binaries, atoms, and
structs. For structs, Phoenix's default behavior is to fetch the id field.
If you have filters with Date, DateTime, NaiveDateTime,
Time values, or any other custom structs (e.g. structs that represent
composite types like a range column), you will need to implement the protocol
for these specific structs in your application.
defimpl Phoenix.Param, for: Date do
  def to_param(%Date{} = d), do: to_string(d)
end

defimpl Phoenix.Param, for: DateTime do
  def to_param(%DateTime{} = dt), do: to_string(dt)
end

defimpl Phoenix.Param, for: NaiveDateTime do
  def to_param(%NaiveDateTime{} = dt), do: to_string(dt)
end

defimpl Phoenix.Param, for: Time do
  def to_param(%Time{} = t), do: to_string(t)
end
It is important that the chosen string representation can be cast back into
the Ecto type.
Examples
iex> to_query(%Flop{})
[]

iex> f = %Flop{page: 5, page_size: 20}
iex> to_query(f)
[page_size: 20, page: 5]

iex> f = %Flop{first: 20, after: "g3QAAAABZAAEbmFtZW0AAAAFQXBwbGU="}
iex> to_query(f)
[first: 20, after: "g3QAAAABZAAEbmFtZW0AAAAFQXBwbGU="]

iex> f = %Flop{
...>   filters: [
...>     %Flop.Filter{field: :name, op: :=~, value: "Mag"},
...>     %Flop.Filter{field: :age, op: :>, value: 25}
...>   ]
...> }
iex> to_query(f)
[
  filters: %{
    0 => %{field: :name, op: :=~, value: "Mag"},
    1 => %{field: :age, op: :>, value: 25}
  }
]
iex> to_query(f)
[filters: %{0 => %{value: "Mag", op: :=~, field: :name}, 1 => %{value: 25, op: :>, field: :age}}]

iex> f = %Flop{page: 5, page_size: 20}
iex> to_query(f, default_limit: 20)
[page: 5]
Encoding the query as a string:
iex> f = %Flop{order_by: [:name, :age], order_directions: [:desc, :asc]}
iex> to_query(f)
[order_directions: [:desc, :asc], order_by: [:name, :age]]
iex> f |> to_query |> Plug.Conn.Query.encode()
"order_directions[]=desc&order_directions[]=asc&order_by[]=name&order_by[]=age"

  


        

      


  

    
Flop.Phoenix.Pagination 
    



      
Defines a struct that holds the information needed to render a pagination
component.

      


      
        Summary


  
    Types
  


    
      
        t()

      


        Describes the data needed to render a pagination component.



    





  
    Functions
  


    
      
        new(meta, opts \\ [])

      


        Returns a Pagination struct for the given Flop.Meta struct.

          @type t() :: %Flop.Phoenix.Pagination{
  current_page: pos_integer() | nil,
  ellipsis_end?: boolean(),
  ellipsis_start?: boolean(),
  next_cursor: String.t() | nil,
  next_direction: :previous | :next,
  next_page: pos_integer() | nil,
  page_range_end: pos_integer() | nil,
  page_range_start: pos_integer() | nil,
  pagination_type: Flop.pagination_type(),
  path_fun:
    (pos_integer() | nil -> String.t())
    | (String.t() | nil, :previous | :next -> String.t()),
  previous_cursor: String.t() | nil,
  previous_direction: :previous | :next,
  previous_page: pos_integer() | nil,
  total_pages: pos_integer() | nil
}


      


Describes the data needed to render a pagination component.
For page-based pagination
	current_page
	ellipsis_end? - Whether an ellipsis should be rendered between the middle
pagination links and the link to the last page.
	ellipsis_end? - Whether an ellipsis should be rendered between the link to
the first page and the middle pagination links.
	next_page
	page_range_start, page_range_end - The range for the links for
individual pages.
	pagination_type - In the case of page-based pagination, this is either
:page or :offset.
	path_fun - 1-arity function that takes a page number and returns a path
to that page that also includes query parameters for filters and sorting.
	previous_page
	total_pages

For cursor-based pagination
	next_cursor - The cursor to be used for the link to the next page.
Depending on the value of the reverse option, this is either the start
cursor or the end cursor of the Flop.Meta struct.
	next_direction - The pagination direction for the link to the next page.
If the reverse option is set to true, this will be :previous.
	pagination_type - In the case of cursor-based pagination, this is either
:first or :last.
	path_fun - 2-arity function that takes a cursor and a direction and
returns a path to that page that also includes query parameters for filters
and sorting.
	previous_cursor - The cursor to be used for the link to the previous page.
Depending on the value of the reverse option, this is either the start
cursor or the end cursor of the Flop.Meta struct.

      new(meta, opts \\ [])



        
          
        

    

  


  

      

          @spec new(
  Flop.Meta.t(),
  keyword()
) :: t()


      


Returns a Pagination struct for the given Flop.Meta struct.
Options
	page_links - Defines how many page links to render. Only used for
page-based pagination. Default: 5.
	path - The path to the current page in the format as accepted by
Flop.Phoenix.build_path/3. Default: nil.
	reverse - Reverses the position of the previous and next link. Only used
for cursor-based pagination. Default: false.


  


        

      


  

    
Flop.Phoenix.InvalidFilterFieldConfigError exception
    



      
Raised when the fields attribute of the Flop.Phoenix.filter_fields
component is set to an invalid value.

      


      
        Summary


  
    Functions
  


    
      
        message(map)

      


        Callback implementation for Exception.message/1.

Callback implementation for Exception.message/1.

  


        

      


  

    
Flop.Phoenix.NoMetaFormError exception
    



      
Raised when a Phoenix.HTML.Form struct is passed to
Flop.Phoenix.filter_fields/1 that was not built with Flop.Meta struct.

    Functions
  


    
      
        message(_)

      


        Callback implementation for Exception.message/1.

      message(_)



        
          
        

    

  


  

Callback implementation for Exception.message/1.

  


        

      


  

    
Flop.Phoenix.PathOrJSError exception
    



      
Raised when a neither the path nor the on_* attribute is set for a
pagination or table component.

      


      
        Summary


  
    Functions
  


    
      
        message(map)

      


        Callback implementation for Exception.message/1.

      message(map)



        
          
        

    

  


  

Callback implementation for Exception.message/1.

